GAS 3.2

G-Engine Auto Scripting language

Version 2 has removed the <relative> parameter from ANIM and ONEOF.

Version 3 added to the WHENNEAR, added DLG to play one line at a time and a new alternative cleanup that will be applied only when transitioning between idle fidgets and the talk / listen set in a conversation.

Version 3.1 added support for the Actor’s facial functions. Note there is also support for a new cleanup: USE CLEARMOOD and USETALK CLEARMOOD. The current version doesn’t support the LookAt and StopLookAt.

Version 3.2 adds placing a limited language set into a prop to cause it to animate. See section entitled GasProps.

ANIM

<filename>,<moving>,<percent>
Plays an animation file specified by <filename>.

<moving> indicates whether or not we should allow a relative animation to move the model. This value is ignored if the animation is absolute. The default value if false

<percent> is optional and is the percent chance that the animation will play when this command is called.
examples:
ANIM
JeanDskFidget01

ANIM
JeanDskFidget02,FALSE

ANIM JeanDskFidget04,FALSE,50

ANIM BucShift1,TRUE

ONEOF
<filename>,<moving>,<percent>
Plays an animation file specified by <filename>. When there are a group of ONEOF commands in a script, each time that group is reached the script will randomly choose one of them to play (hence the name)

<moving> indicates whether or not we should allow a relative animation to move the model. This value is ignored if the animation is absolute. The default value if false

<percent> is optional and is the percent chance that the animation will play when this command is called.
examples:
ONEOF
JeanDskFidget01,FALSE

ONEOF
JeanDskFidget02,FALSE,50

ONEOF
BucShift1,TRUE,TRUE

WAIT

<minimum>,<maximum>,<percent>

Waits (i.e. does nothing) for a number of seconds

<minimum> is the minimum number of seconds to wait

<maximum> is optional and is the maximum number of seconds to wait. If this value is 0 or not present then the <minimum> is used.

<percent> is optional and is the percent chance that the script will wait play when this command is called.
examples:
WAIT 1

WAIT 1,5

WAIT 1,5,75

WAIT 2,0,50
LABEL
<labelname>

Creates a label called <labelname> which GOTO and IF commands can jump to.

examples:
LABEL
foo
GOTO
<labelname>

Jumps to <labelname>

examples:
GOTO

foo
LOOP

Jumps to the beginning of the file

examples:
LOOP
SET
<var> <value>

Sets a variable called <var> to <value>

Variable names should only be a single character such as X or Y

examples:
SET
X
0

SET
Y
7

INC
<var>

Adds one to <var>

Variable names should only be a single character such as X or Y

examples:
INC
X

DEC
<var>

Subtracts one from <var>

Variable names should only be a single character such as X or Y

examples:
DEC
X

IF
<var> <compare> <value> <label>

<var> is a variable to check

<compare> is a comparison operator (i.e. >,=, or <)

<value> is a value to compare var to

<label> is the label to jump to if the comparison is true

Variable names should only be a single character such as X or Y

examples:

IF
X > 5
foo

IF
Y = 5
bar

IF
Z < 5
SecondLoop

Sample Script:
ANIM JeaLbyArmsUp2Dsk

SET X 0

LABEL One

ONEOF JeaLbyReachDwnDsk

ONEOF JeaLbyArmsDwnDsk

INC X

IF X > 2 Two

GOTO One

LABEL Two

ANIM JeaLbyTapDskRgstr

LOOP

Additions to GAS version 2:

The additions use position and model information. The .SIF file for a given scene determines what positions and models are available for use.

WALKTO   <position>

WALKTO   <x, y, z>

Requires that an IPOS has been indicated. See next.

<position> is the position in the room to walk to.

<x, y, z> are the coordinates in the room to walk to.

examples:

WALKTO Hall_Door

CHOOSEWALK
 (  <position>,<position>,<position>, …  )

This simplifies wandering randomly between a set of three or more positions. There is not an upper limit on the number of positions that can be in the set. 

When choosing which position to walk to next the current position is excluded from the set. 

Requires that an IPOS has been indicated and a space is used around each of the parenthesis.

Each <position> is one of a set of room positions.

examples:

CHOOSEWALK ( Start, stairs_right, picture )

Sample ChooseWalk Script:

USE IPOS start

ANIM BuchScratch

LABEL start_rounds

CHOOSEWALK ( stairs_right, picture, start )

ANIM BuchLook

ANIM BuchScratch

GOTO start_rounds


This example behaves exactly like the previous one except the order of walking will not always be stairs_right – picture – start – stairs_right … 

USE IPOS   <position>

IPOS is the “interrupt position”. Allowing model movement with walking requires that we have a position given where any interaction with Ego is to occur. 

<position> is the position in the room from which ego interaction happens.

examples:

USE IPOS Start

Sample WalkTo Script:
USE IPOS start

ANIM BuchScratch

LABEL start_rounds

WALKTO stairs_right

ANIM BuchLook

ANIM BuchScratch

WALKTO picture

ANIM BuchLook

ANIM BuchScratch

WALKTO start

ANIM BuchLook

ANIM BuchScratch

GOTO start_rounds


Buchelli will wander between the “stairs_right” and “picture” scene positions, animating a little at each. When Ego clicks Talk on Buchelli, for example, Buchelli will walk to scene position “start” before the Talk sequence is started.

USE CLEANUP <filename1>, <filename2>
USE NEWIDLE <filename.gas>

If when an animation completes it does not leave the model in its default pose we say that animation has a dependency. It depends on another animation to return to its default. USE CLEANUP allows unwinding such hierarchies as quickly as possible.

<filename1> is the .anm file that doesn’t end in default

<filename2> is the .anm that should be played after the first one to get the model to or towards its default pose.

Just ask if the character is interrupted while playing filename1 what should be played next to move towards the character’s default ?

examples:


The first two examples assume that after playing putDownPaper the character has returned to their default pose.


// One level dependency

USE CLEANUP pickupPaper, putDownPaper

// Two level dependency

USE CLEANUP pickupPaper, putDownPaper

USE CLEANUP liftPaper, lowerPaper

USE CLEANUP lowerPaper, putDownPaper



// Etc. – After “MosLeanB” Mosely is in default.

USE CLEANUP MosLeanA, MosLeanB

USE CLEANUP MosArmJumpA, MosArmJumpB

USE CLEANUP MosJumpingA, MosJumpingB

USE CLEANUP MosJumpingB, MosArmJumpB

USE CLEANUP MosArmJumpB, MosLeanB

Sample dependant animations Script:
Madeline is in the kitchen. She scratches her head, reaches up and opens a cupboard. She looks around then reaches in to take a coffee can. Shakes the can while listening then puts it away and closes the cupboard. “MadDown” leaves her in default.

USE CLEANUP MadReachUp, MadDown

USE CLEANUP MadOpenCupboard, MadCloseCupboard

USE CLEANUP MadCloseCupboard, MadDown

USE CLEANUP MadLookUp, MadCloseCupboard

USE CLEANUP MadGetCan, MadPutawayCan

USE CLEANUP MadShakeCan, MadPutawayCan

USE CLEANUP MadPutawayCan, MadCloseCupboard

ANIM MadReachUp

ANIM MadOpenCupboard

ANIM MadLookUp

ANIM MadGetCan

ANIM MadPutawayCan

ANIM MadGetCan

ANIM MadShakeCan

ANIM MadPutawayCan

ANIM MadCloseCupboard

ANIM MadDown


Now Gabe can interrupt Madeline at any point and she will have returned to her default within (at most) three animations.

Use Form 2:

USETALK CLEANUP <filename1>, <filename2>

USETALK IPOS  <position>

USETALK NEWIDLE <filename.gas>


By utilizing the keyword USETALK the exact cleanup functionality is available that has been described above. The difference is that this cleanup alternative is used, if supplied, whenever a conversation is entered. A conversation begins by transitioning from a character’s idle fidget into their talk / listen set. Cleanup is applied when leaving the idle. 


An example of its use is where Gabe is talking to Mosely in his room. They always converse while sitting down. Gabe’s idle on the bed includes a USE CLEANUP set that properly stands him up in response to any user click that causes Gabe to walk. The USETALK form addresses the issue that if the user click causes a conversation we do not want Gabe to stand up. This form of NEWIDLE allows switching between standing and the sitting idle gas files. 


Both forms of cleanup exist in the same file.

Sample dual cleanup:


If interrupted for a conversation she might leave the cupboard open. For all other interruptions she will close the cupboard:

USE CLEANUP MadReachUp, MadDown

USE CLEANUP MadOpenCupboard, MadCloseCupboard

USE CLEANUP MadCloseCupboard, MadDown

USETALK CLEANUP MadReachUp, MadDown

USETALK CLEANUP MadOpenCupboard, MadDown

USETALK CLEANUP MadCloseCupboard, MadDown

USE CLEARMOOD

USETALK CLEARMOOD

This can be used to assure that any Gas that has set a mood will clear that mood whenever the Actor is called to attention. It is applied as the last step of the cleanup.

WHENNEAR <Noun>, <distance>, <Label>
This behaves like a trigger.  When the model indicated by the noun is within the given distance the gas file will go to the indicated label as soon as possible.

An example of it’s use is when Emilio is sitting on the bench outside the hotel reading a newspaper (of course we know he is spying on Gabe…) Whenever Gabe approaches to talk with him Emilio is to get up and return to the hotel lobby.

Sample WhenNear Script:

LABEL PaperFidget

WHENNEAR Gabriel, 30, EnterHotel

ANIM EmlScratch

ANIM EmlPaperShake

ANIM EmlLookOverPaper

GOTO PaperFidget

LABEL EnterHotel

ANIM EmlStand

WALKTO ToHotel

ANIM EmlEnterHotel

example:

WHENNEAR GABRIEL, 30, ENTERHOTEL

Form 2:

WHENNEAR <Noun>, <distance>, <Label>, <Noun>
This behaves like a trigger.  When the 1st noun model is within the given distance from the second noun model the gas file will go to the indicated label as soon as possible. The models can be scene models, props or actors.

An example of it’s use is when Jean is wandering around the Lobby cleaning. Whenever Gabe approaches the desk Jean needs to return to his place behind it.

Sample WhenNear Script:

LABEL Start

WHENNEAR Gabriel, 10, ReturnToDesk, LbyDesk

WALKTO (cleanPlace1)

ANIM Clean1

WALKTO (cleanPlace2)

ANIM Clean2

GOTO Start

LABEL ReturnToDesk

WALKTO BehindDeskPos

example:

WHENNEAR GABRIEL, 10, RETURNTODESK, LBYDESK

NEWIDLE <filename.gas>
This allows chaining gas scripts together. It will do whatever cleanup is needed to assure the model is in default position and then end the current script and start the indicated one.

example:

NEWIDLE  MadNext.gas

DLG <filename>

This statement allows a GAS file to play a single dialogue (yak file) line at a time. It resolves to a StartDialogue call similar to the sheep function. An example of when this will be useful in conjunction with the WHENNEAR for a quick voice over line addressing the ego as it walks by.


This is not meant for extended dialogue comprising many lines.


Do not include the .yak extension.


Do not include the language pre-fix!! This will just make localizing GAS files a lot of work.

example:

DLG  014ED3SLK1

WHENNOLONGERNEAR <Noun>, <distance>, <Label>
This behaves like a trigger.  When the model indicated by the noun is not within the given distance the gas file will go to the indicated label as soon as possible.

An example of this is when Jean and Gabe have finished their conversation in the hotel lobby.  Jean does not continue walking around the Lobby until Gabe has walked away from him.  So we will set up a  WhenNoLongerNear script to determine when Gabe has walked away.

Sample WhenNoLongerNear Script:

WHENNOLONGERNEAR Gabriel, 30, ResumeWalk

ANIM jeaFig1

ANIM jeaFig2

ANIM jeaFig3

LOOP

LABEL ResumeWalk

NEWIDLE jeaResWalkIdle.gas

Form 2:

WHENNOLONGERNEAR <Noun>, <distance>, <Label>, <Noun>
This behaves like a trigger.  When the 1st noun model is not within the given distance from the second noun model the gas file will go to the indicated label as soon as possible. The models can be scene models, props or actors.

example:

WHENNOLONGERNEAR GABRIEL, 100, EstelleResumeIdle, LADY_HOWARD

WHENINVIEW <Noun>, <FOV>, <Label>
This behaves like a trigger.  When the model indicated by the noun is within the actor’s line of sight  given by the FOV, the gas file will go to the indicated label as soon as possible.

An example is when Mosely sees Grace after she exits from CdS Ext.

Sample WhenInView Script:

WHENINVIEW Grace, 70, SayHi

ANIM MosFig1

ANIM MosFig1

ANIM MosFig1

LOOP

LABEL SayHi

DLG 18005551234

LOOP

example:

WHENINVIEW GRACE, 30, SAYHI

LOCATION <location>
T his will set the actor’s location to location.  This is the same as calling SetActorLocation(“ActorRunningTheGas”,”location”);

Sample Location Script:

WHENNEAR Gabriel, 30, EnterHotel

ANIM EmlSitFig2

ANIM EmlBreath

ANIM EmlSitFig1

LOOP

LABEL EnterHotel

ANIM EmlStand

LOCATION LBY

WALKTO ToHotel

ANIM EmlEnterHotel

example:

LOCATION LBY

SETMOOD <mood>
This will set the actor’s mood. Note this stays set until cleared.

example:

SETMOOD ANGRY

CLEARMOOD

This will clear the actor’s mood. 

example:

CLEARMOOD

GLANCE <percentX,  percentY, duration>

This will cause an actor to glance for the given duration. See Sheep.Doc for details.

example:

GLANCE 3,4,8
// 3%, 4%, for 8 Ms

GLANCE <percentLeftX, percentLeftY, percentRightX, percentRightY , duration>

This will cause an actor to glance for the given duration. See Sheep.Doc for details.

example:

GLANCE 3,4,5,6,8
// Left:3%x, 4%y  Right:5%x, 6%y, for 8 Ms

TURNHEAD <percentX,  percentY, duration>

This will cause an actor to turn their head the given percentages for the given duration. See Sheep.Doc for details. Default duration is 1000 ms or 1 second.

example:

TURNHEAD 12, 20, 8
// 12% on X, 20% on Y, for 8 Ms

LOOKAT <noun,  spec, duration>

This will cause an actor to look at the given noun for the given duration. The spec is E = eyes, H = head. The sheep version supports B = body and A = All but we do not want to use the B in Gas so an error is written if the spec includes an A or a B. See Sheep.Doc for details. Default duration is 2 seconds.

example:

LOOKAT Plant EH 5
// lookAt the plant with eyes and head // for 5 seconds.
EXPRESSION <expression>
This will set the actor’s expression. This is a one shot emotion.

example:

EXPRESSION ANGRY

GasProps

These are non-actor models with gas files. They are created in the models section of the .sif file with a type=gasprop and optionally a gas=xxx.gas parameter. Many of the Gas statements only make sense when used with an actor so a subset is supported for props. The statements that can be used in props are:

· ANIM

· ONE OF

· WAIT

· LOOP

· IF

· GOTO

· LABEL

· SET

· INC

· DEC

· //
(comments)

These next two only make sense if used in the form of sending in two nouns:

· WHENNEAR

· WHENNOLONGERNEAR

